A színpadra eljárással hat objektumot hozok létre (öt teknőc, egy nyomógombot).

to színpadra

Newturtle "Ádám setc 116 st pd

Newturtle "Éva setc 16 st pd

Newturtle "Káin st pd

Newturtle "Lucifer setc 45

newturtle "Kígyó setc 65

newbutton "sz [-220 180][everyone [setpos [0 0] seth 0] clean]

end

A felkészül metódusokat rendel a főszereplőkhöz. Lucifer metódusa változó, a  Kígyóé pedig nyilvános.

to felkészül

Set "Ádám "onclick [forever [run :a]]

Set "Éva "onclick [forever [run :b]]

Set "Káin "onclick [forever [run :a run :b]]

set "sz "label "szivacs

end

Egyenesvonalú mozgások szuperpoziciója

to egyenes

make "a [setx xcor + 1]

make "b [sety ycor + 1]

ask [Ádám Éva Káin] [clickon]

end

to kör

make "a [fd 1]

make "b [rt 1]

everyone [clickon]

end

Az ordináta változása korlátok között

to fűrészfog

make "a [setx xcor + 0,5]

make "b [fd 1 if or ycor > 80 ycor < -80 [rt 180]]

ask [Éva Káin][seth 0]

ask [Ádám Éva Káin] [clickon]

end

Az ordináta harmonikusan változik

to szinusz

;a színpadon már ott a három főszereplő: Ádám, Éva, Káin, valamint Lucifer

Lucifer, st pd setpos [-57 0] seth 0

set "Lucifer "onclick [forever [fd 1 rt 1]]
make "a [setx xcor + 1]

make "b [sety ask "Lucifer [ycor]]

everyone [clickon]

end

Koszinusz a vízszintes metszék:

to koszinusz

;a színpadon már ott a három főszereplő: Ádám, Éva, Káin, valamint Lucifer

Ádám, setx -120

Lucifer, st pd setpos [57 0] seth 0

set "Lucifer "onclick [forever [fd 1 lt 1]]
make "a [setx xcor + 1]

make "b [sety ask "Lucifer [xcor]]

everyone [clickon]

end

A szinuszgörbét áthelyezem

to koszinusz2

set "Lucifer "onclick [forever [fd 1 rt 1]]

Lucifer, st pd setpos [0 57]seth 90

make "a [setx xcor + 1]

make "b [sety ask "Lucifer [ycor]]

everyone [clickon]

end

Egy teknőcöt a másikhoz kötök

to köt :a

 ;kötöm az aktuális teknőcöt az „a” teknőchöz

 towards :a

 make "d distance :a

 make "alfa ask :a [heading]

 make "x ask :a [xcor]

 make "y ask :a [ycor]

 setpos list :x + :d * sin :alfa :y + :d * cos :alfa 

end

A forgó teknőchöz kötött teknőc kering

to forgat

make "a [rt 1]

make "b [köt "Ádám]

ask [Ádám Éva] [clickon]

end

A keringetett teknőc önálló mozgást is végez, sugárirányban is halad

to spiral

make "a [rt 1]

make "b [köt "Ádám bk 0.5]

ask [Ádám Éva] [clickon]

end

A keringetett teknőc nem hagyja el az y tengellyel párhuzamos egyenest.

tgjelenet a főperiodus generálását mutatja

to tgfp

Ádám, pu sety 200 setx -300 pd

Éva, pu setx 210  

Káin, st pu setx -300 pd 

Lucifer, setc 45 seth 160 st

Kígyó, st

set "Ádám "onclick [forever [setx xcor + 0.025]]

set "Éva "onclick [forever [sety ask "Kígyó [ycor]]]

set "Káin "onclick [forever [setx ask "Ádám [xcor] sety ask "Éva [ycor]]]

set "Lucifer "onclick [forever [ifelse heading > 13 [lt 0,05] [seth 166]]]

set "Kígyó "onclick [forever [köt "Lucifer setx 57]]

everyone [clickon]

end
tangens és kotangens

to tg_ctg

Ádám, pu sety 200 setx -300

Éva, pu setx 210  

Káin, st setx -300 pd 

Lucifer, setc 45 seth 160 st

Kígyó, st pu

set "Ádám "onclick [forever [setx xcor + 0.025]]

set "Éva "onclick [forever [sety ask "Kígyó [ycor]]]

set "Káin "onclick [forever [setx ask "Ádám [xcor] sety ask "Éva [ycor]]]

set "Lucifer "onclick [forever [lt ,05]]

set "Kígyó "onclick [forever [köt "Lucifer setx 57]]

everyone [clickon]

end

Éva az élbelátszó síkra vetített lézerpontot követi

to köt2 :a

 ;kötöm az aktuális teknőcöt az „a” teknőchöz

 towards :a

 make "d distance :a

 make "alfa ask :a [heading]

 make "x ask :a [xcor]

 make "y ask :a [ycor]

 setpos list :x - :d * sin :alfa :y - :d * cos :alfa 

end
to tangenshez

Lucifer, st pd setx 0 seth 90

set "Lucifer "onclick [forever [lt 0,025]]

Kígyó, st setpos [0 0]

Ádám, pu setpos [-200 180]

Éva, pu setx -280 seth 0

Káin, st pu setx -300 

make "a [setx xcor + 0,05]

make "b [sety ask "Kígyó [ycor]]

ask [Ádám Éva Káin Lucifer][clickon]

end

to tg

make "k ask "Káin [ycor]

make "l1 (ask "Lucifer [heading]) < 165

make "l2 (ask "Lucifer [heading]) > 15

make "ji and :l1 :l2

make "l3 (ask "Lucifer [heading]) < 345

make "l4 (ask "Lucifer [heading]) > 195

make "bi and :l3 :l4

if :ji [ifelse :k < -170 [Káin, pu][Káin, pd] Kígyó, köt "Lucifer setx 57 ]

ifelse not :k < 169  [Káin, pu make "a [setx xcor + 0,01]] [make "a [setx xcor + 0,05]]

if :bi [ifelse :k < -169 [Káin, pu][Káin, pd] Kígyó, köt2 "Lucifer setx 57]

tg

end
to tangens

tangenshez

tg

end

